Homework No. 1

Jorge A. Ramirez

Define constants

```
> Lv:=2500000;Rv:=461.5;Cp:=1004;Rd:=287;g:=9.81;rhow:=1000;
  epsilon:=0.622;
```

$$Lv := 2500000$$
$$Rv := 461.5$$
$$Cp := 1004$$
$$Rd := 287$$
$$g := 9.81$$
$$rhow := 1000$$
$$\varepsilon := 0.622 \qquad (1)$$

Define potential temperature equation.

```
> theta:=(T,p)->T*(100000/p)^(Rd/Cp);
```

$$\theta := (T, p) \rightarrow T\left(\frac{100000}{p}\right)^{\left(\frac{Rd}{Cp}\right)} \qquad (2)$$

Ideal Gas Law.

```
> rho:=(p,T,R)->p/R/T;
```

$$\rho := (p, T, R) \rightarrow \frac{p}{R\,T} \qquad (3)$$

Poisson equation for pressure following a dry adiabatic process

```
> p2:=(T,thetao)->10^5*(T/thetao)^(Cp/Rd);
```

$$p2 := (T, thetao) \rightarrow 100000\left(\frac{T}{thetao}\right)^{\left(\frac{Cp}{Rd}\right)} \qquad (4)$$

Clausius-Clapeyron equation for saturation vapor pressure

```
> es:=(T)->611*exp(-Lv/Rv*(1/T-1/273.15));
```

$$es := T \rightarrow 611\,\mathrm{e}^{\left(-\frac{Lv\left(\frac{1}{T}-\frac{1}{273.15}\right)}{Rv}\right)} \qquad (5)$$

Saturation mixing ratio

```
> ws:=(T,p)->0.622*es(T)/p(T,thetao);
```

$$ws := (T, p) \rightarrow \frac{0.622\,es(T)}{p(T, thetao)} \qquad (6)$$

Homework No. 1

1. At a climate station, the following measurements are made: air pressure = 1011.0 hPa, air temperature = 25 ∫C, and dew point temperature = 20 ∫C. Calculate the corresponding vapor pressure, relative humidity, specific humidity, and air density.

```
> po:=101100; To:=273.15+25; Td:=273.15+20;
```

$$po := 101100 \quad (7)$$
$$To := 298.15$$
$$Td := 293.15$$

Compute actual vapor pressure. Use definition of dew point temperature to do so.

```
> e:=es(Td);
```

$$e := 2364.037507 \quad (8)$$

Compute Relative Humidity:

```
> rh:=e/es(To)*100;
```

$$rh := 73.35244849 \quad (9)$$

Compute water vapor mixing ratio:

```
> w:=0.622*e/(po-e);
```

$$w := 0.01489256085 \quad (10)$$

```
> w:=0.622*e/po;
```

$$w := 0.01454432571 \quad (11)$$

Compute* Specific* Humidity:

```
> q[h]:=w/(1+w);
```

$$q_h := 0.01433582086 \quad (12)$$

Compute density of moist air,

$$\rho_m := \frac{po\left(1 - \frac{0.378\,e}{po}\right)}{Rd\;To} \quad (13)$$

```
> rho[m]:=po/Rd/To*(1-0.378*e/po);
```

$$\rho_m := 1.171058857 \quad (14)$$

2. A sample of moist air has a temperature of 280 ∫K at a pressure of 900 hPa, with a mixing ratio of 5g/kg. Compute the following quantities for this sample: a) virtual temperature; b) absolute humidity; c) specific humidity; d) relative humidity; e) dew point temperature; f) potential temperature; and g) equivalent potential temperature.

```
> To:=280; po:=90000; w:=0.005;
```

$$To := 280 \quad (15)$$
$$po := 90000$$
$$w := 0.005$$

Virtual Temperature:

$$Tv := \frac{T}{1 - \frac{(1-\varepsilon)\,e}{p}} = \frac{T}{1 - \frac{(1-\varepsilon)\,w}{\varepsilon}} \quad (16)$$

```
> Tv:=To/(1-(1-epsilon)*w/epsilon);
```

$$Tv := 280.8533970 \quad (17)$$

To compute absolute humidity we use the ideal gas law applied to the water vapor. However, we need to obtain the actual water vapor pressure first as $e := \frac{w\, p_o}{\varepsilon}$ and $\rho_v := \frac{e}{To\, Rv}$ (18)

```
> e:=w*po/epsilon; rho[v]:=e/To/Rv;
```

$$e := 723.4726688$$
$$\rho_v := 0.005598766977 \quad (19)$$

Specific Humidity:

```
> q[h]:=w/(1+w);
```

$$q_h := 0.004975124378 \quad (20)$$

Relative Humidity:

```
> rh:=e/es(To)*100;
```

$$rh := 72.89059549 \quad (21)$$

Compute Dew Point temperature. Solve by trial and error first:

```
> Td:=fsolve(es(T)-e,T, 273.15..To);
```

$$Td := 275.4971936 \quad (22)$$

Obtain Td by explicitly solving for it:

$$Td := T \rightarrow \frac{1}{\frac{1}{273.15} - \frac{Rv \ln\left(\frac{1}{611} \frac{w\, po}{\varepsilon}\right)}{Lv}} \quad (23)$$

```
> Td:=1/(1/273.15-Rv/Lv*ln(w*po/611/epsilon));
```

$$Td := 275.4971936 \quad (24)$$

Compute Potential Temperature using potential temperature equation defined above:

```
> thetao:=evalf(theta(To,po));
```

$$thetao := 288.5613165 \quad (25)$$

Parcel temperature at the Lifting Condensation Level following an adiabatic ascent. At the LCL, the initial mixing ratio is equal to the saturation mixing ratio at the pressure of the LCL and at a parcel temperature corresponding to an adiabatic ascent.Solve using trial and error.

```
> Ts:=fsolve(ws(TT,p2(TT,thetao))-w=0,TT,273.15..To);
```

$$Ts := 274.5343525 \quad (26)$$

Parcel pressure at the Lifting Condensation Level following an adiabatic ascent. Obtain ps using the Poisson equation defined above.

```
> ps:=p2(Ts,thetao);
```

$$ps := 84002.60411 \quad (27)$$

Compute equivalent potential temperature as:

$$\theta_e := (\theta, T, p) \rightarrow \theta\, e^{\left(\frac{Lv\, ws(T, p)}{Cp\, T}\right)} \quad (28)$$

Define equivalent potential temperature equation:

```
> thetae:=(thetat,TT,pp)->thetat*exp(Lv*ws(TT,pp)/Cp/TT);
```

$$thetae := (thetat, TT, pp) \rightarrow thetat\, e^{\left(\frac{Lv\, ws(TT, pp)}{Cp\, TT}\right)} \quad (29)$$

Evaluate equivalent potential temperature equation:

```
> thetae(thetao,Ts,ps);
```

$$301.9489127 \quad (30)$$

3. Derive expressions for the vertical distributions of pressure, p(z), and density, $\rho(z)$, of an atmosphere whose temperature decreases linearly with elevation at a rate given by Γ. Graph your results.

Vertical pressure distribution in a hydrostatic atmosphere:

$$\frac{d}{dz} p(z) = -\rho(z)\, g$$

Use ideal gas law

$$\rho(z) := \frac{p(z)}{R\, T(z)}$$

and substitute in hydrostatic equation to obtain:

$$\frac{d}{dz} p(z) = -\frac{g\, p(z)}{R\, T(z)}$$

Integrating this equation yields:

$$\ln(p(z)) := -\ln(p(z_o)) - \frac{g \int_{z_o}^{z} \frac{1}{T(z)}\, dz}{R}$$

However, in order to integrate the right hand side we must know the variation of ambient temperature with elevation, which is given by:

$$T(z) := T_o - \Gamma\,(z - z_o)$$

Substituting and integrating obtain:

$$\ln(p(z)) := -\ln(p(z_o)) - \frac{g \int_{z_o}^{z} \frac{1}{T_o - \Gamma\,(z - z_o)}\, dz}{R}$$

$$p(z) := p(z_o)\left(\frac{T_o - \Gamma\,(z - z_o)}{T(z_o)}\right)^{\left(\frac{g}{R\,\Gamma}\right)}$$

The above result is the vertical variation of pressure in a hydrostatic atmosphere and for a linear temperature variation of temperature with elevation.

Using the ideal gas law to substitute density for pressure obtain:

$$\rho(z) := \rho(z_o)\left(\frac{T_o - \Gamma(z - z_o)}{T(z_o)}\right)^{\left(\frac{g}{R\Gamma} - 1\right)} \tag{31}$$

These results can be graphed in dimensionless form for a given lapse rate, for example, 0.006 K/m, as follows:

```
> To:=300;Rd:=287;g:=9.81;Gamma:=0.006;zo:=0;
```

$$To := 300$$
$$Rd := 287$$
$$g := 9.81$$
$$\Gamma := 0.006$$
$$zo := 0 \tag{32}$$

```
> rho:=z->((To-Gamma*(z-zo))/To)^(g/Rd/Gamma-1);
```

$$\rho := z \rightarrow \left(\frac{To - \Gamma\,(z - zo)}{To}\right)^{\left(\frac{g}{Rd\,\Gamma} - 1\right)} \tag{33}$$

```
> p:=z->((To-Gamma*(z-zo))/To)^(g/Rd/Gamma);
```

$$p := z \rightarrow \left(\frac{To - \Gamma\,(z - zo)}{To}\right)^{\left(\frac{g}{Rd\,\Gamma}\right)} \tag{34}$$

```
> plot([p(z), rho(z)],z=0..50000);
```



4. Moist air at 1000 hPa and 25 ∫C has a wet-bulb temperature of 20 ∫C. Find the dew point temperature. If this moist air were expanded until all the moisture condensed and fell out and then compressed to 1000 hPa, what would be the resulting temperature? Assume this process is adiabatic while the parcel is unsaturated, and moist adiabatic while the parcel is saturated.

From the definition of wet bulb temperature, and the definition of dew point temperature we can obtain the following equation:

$$wo := ws(Tw, p) + \frac{Cp\ (Tw - T)}{Lv} = \frac{\varepsilon\, es(Td)}{p} \tag{35}$$

which can be solved for Td as indicated below.

```
> po:=100000;Tw:=273.15+20;To:=273.15+25;
```

$$po := 100000$$
$$Tw := 293.15$$
$$To := 298.15 \tag{36}$$

Compute dew point temperature:

```
> Td:=fsolve(epsilon*es(Tw)/po+Cp*(Tw-To)/Lv-epsilon*es(T)/po, T);
```

$$Td := 290.8390699 \tag{37}$$

Compute potential temperature:

```
> thetaoo:=evalf(theta(To,po));
```

$$thetaoo := 298.15 \tag{38}$$

```
> evalf(theta(298.15,100000));
```

(39)

$$298.15 \tag{39}$$

```
> epsilon*es(Td)/po;
```

$$0.01269631326 \tag{40}$$

Finally, we are asked to compute the equivalent potential temperature:

$$\theta_e := \theta\, e^{\left(\frac{L_v w_s(T_s, p_s)}{C_p T_s}\right)} \tag{41}$$

Thus, we need to solve for Ts and ps, as done in Problem 2 above.

```
> thetao:=evalf(theta(To,po));
```

$$thetao := 298.15 \tag{42}$$

```
> Ts:=fsolve(ws(TT,p2(TT,thetao))-epsilon*es(Td)/po,TT,289..290);
```

$$Ts := 289.1799353 \tag{43}$$

```
> ps:=p2(Ts,thetao);
```

$$ps := 89864.84127 \tag{44}$$

```
> thetae:=(thetat,TT,pp)->thetat*exp(Lv*ws(TT,pp)/Cp/TT);
```

$$thetae := (thetat, TT, pp) \rightarrow thetat\, e^{\left(\frac{Lv\, ws(TT, pp)}{Cp\, TT}\right)} \tag{45}$$

```
> thetae(thetao,Ts,ps);
```

$$332.5934189 \tag{46}$$

5. The equivalent potential temperature of an unsaturated parcel of surface air is 340 ∫K, and its potential temperature is 300 ∫K. If the ambient air temperature at the lifting condensation level is 295 ∫K, compute the buoyancy force acting on the parcel at that level. Assume the surface is at a pressure of 1000 hPa. State all your assumptions.

$$Fb := g\left(\frac{\rho_a}{\rho_p} - 1\right) = g\left(\frac{Tp}{Ta} - 1\right)$$

In order to compute the buoyancy force acting on the parcel at the LCL using the above equation, we must find the temperature of the parcel at the LCL. Assume that the ascent to LCL occurs adiabatically. Thus, potential temperature and mixing ratio are conserved.

At the LCL the following equation holds:

$$w_o := w_s(T_s, p_s) = \frac{\varepsilon\, e_s(T_s, p_s)}{p_s}$$

But, we do not know wo, nor Ts, nor ps. Thus, we need 2 additional equations. However, we know the equivalent potential temperature, which we can use as one of the 2 equations:

$$\theta_e := \theta\, e^{\left(\frac{L_v w_s(T_s, p_s)}{C_p T_s}\right)}$$

The final equation is the potential temperature equation:

$$\theta := T_s \left(\frac{100000 \cdot 1}{p_s} \right)^{\left(\frac{R}{C_p} \right)}$$

Putting these equations together leads to:

$$\theta_e := \theta\, e^{\left(\frac{L_v \varepsilon e_s(T_s)}{p_s C_p T_s} \right)} = \theta\, e^{\left(\frac{1 L_v \varepsilon e_s(T_s)}{100000 \left(\frac{T_s}{\theta} \right)^{\left(\frac{C_p}{R} \right)} C_p T_s} \right)} \tag{47}$$

In the above equation, potential temperature and equivalent potential temperature are known. The only unknown quantity is the parcel temperature at the LCL, which is what we need. We can solve this equation by trial and error. This is done below:

```
> theta:=300;Ta:=295;
```

$$\theta := 300 \tag{48}$$
$$Ta := 295$$

```
> thetae:=Ts->theta*exp(Lv*epsilon*es(Ts)/((Ts/theta)^(Cp/Rd)*10^5)
  /Cp/Ts);
```

$$thetae := Ts \rightarrow \theta\, e^{\left(\frac{1}{100000} \frac{Lv\, \varepsilon\, es(Ts)}{\left(\frac{Ts}{\theta} \right)^{\left(\frac{Cp}{Rd} \right)} Cp\, Ts} \right)} \tag{49}$$

```
> Ts:=fsolve(thetae(Tss)-340,Tss,290..300);Ta:=295;
```

$$Ts := 291.5106111 \tag{50}$$
$$Ta := 295$$

Buoyancy force per unit mass of parcel in N/kg:

```
> F[B]:=g*(Ts/Ta-1);
```

$$F_B := -.1160369670 \tag{51}$$

```
>
```